


花蓮高商程式語言進階

## 第二站第1題：身分證字號檢查

# 開新專案



# 設計版面

身分證號碼檢查

應檢人資料

姓名  准考證號碼

座號  考試日期

## ✕ 表單

- + Form1.text=身分證號碼檢查
- + Form1.size=500,500

## ✕ 容器

- + Groupbox.text=應檢人資料

## ✕ 通用控制項

- + Label1.text=姓名
- + Label2.text=座號
- + Label3.text=術科測試編號
- + Label4.text=考試日期
- + Textbox \*4

## ✕ 資料

- + DataGridView.name=dgv
- + DataGridView.size=450,300

# 主程式架構

```
Form1.vb x Form1.vb [設計]
(一般)
Public Class gb_student
    Dim rno, ec, d(100, 4)
    Private Sub gb_student_Load(sender As System.Object, e As System.EventArgs) Handles MyBase.Load
        Call rdata()
        For i = 1 To rno
            ec = ""
            If ec = "" Then Call sp1(i)
            If ec = "" Then Call sp2(i)
            If ec = "" Then Call sp3(i)
            d(i, 4) = ec
        Next
        Call wdata()
    End Sub
    Sub rdata()

    End Sub
    Sub sp1(ByVal i)

    End Sub
    Sub sp2(ByVal i)

    End Sub
    Sub sp3(ByVal i)

    End Sub
    Sub wdata()

    End Sub
End Class
```

# 邏輯架構

## ✕ 全域變數：

- + rno：每一筆資料之編號

- + 二維陣列： $d(100,4)$

  - ✕ 存放最多101筆資料(第0筆不使用)

  - ✕ 以及每筆資料有5個狀態存放空間

    - ✕  $d(i,0)$ ：不使用， $d(i,1)$ ：放身分證號碼， $d(i,2)$ ：放姓名

    - ✕  $d(i,3)$ ：放性別(M表男性，F表女性)

    - ✕  $d(i,4)$ ：放錯誤代碼ec的值

- + ec：錯誤代碼ErrorCode

  - ✕ FORMAT ERROR

  - ✕ SEX CODE ERROR

  - ✕ CHECK SUM ERROR



## 邏輯步驟

- ✖ 從指定檔案開檔並讀取檔案全部資料
- ✖ 對每一筆資料進行3項檢查
  - + ec值先清空
  - + 格式檢查，通過者進行下1項檢查，否則讓ec=FORMAT ERROR
  - + 性別碼檢查，通過者進行下1項檢查，否則讓ec=SEX CODE ERROR
  - + 檢查碼檢查，未通過者讓ec=CHECK SUM ERROR
- ✖ 將全部的資料內容寫入dgv並排序

## 副程式：rdata()

```
Sub rdata()  
  FileOpen(1, "c:\11900\1060306.sm", OpenMode.Input)  
  rno = 0  
  Do While Not EOF(1)  
    rno += 1  
    For j = 1 To 3  
      Input(1, d(rno, j))  
    Next  
  Loop  
  FileClose(1)  
End Sub
```

'資料編號初始為0  
'當不到檔案結尾時，做loop  
'資料編號在每個loop開始時累加  
'為每筆資料的前3欄位放入讀到的資料

✖ EOF：end of file(表示檔案結尾)

# 格式檢查：sp1(i)

2. 身分證號碼檢查原則如下：

(A) 格式：共有十位，第一位為大寫字母，後九位為數字。表示如下：

L1	D1	D2	D3	D4	D5	D6	D7	D8	D9
----	----	----	----	----	----	----	----	----	----

## × 變數

+ idno(身分證號)

+ m1(驗證長度，使用len()函數)

+ m2(驗證第1碼是否介於A~Z之間)

+ m3(驗證第2~10碼，是否介於0~9之間)



# 副程式sp1(i)的程式碼

```

Sub sp1(ByVal i)
    Dim idno, m1, m2, m3          '宣告idno(放身分證字號),m1(放身分證長度),m2(放身分證第1碼),m3(放身分證第2~第10碼)
    idno = d(i, 1)                '宣告idno，並指派第i筆資料的第1個欄位(身分證字號)之內容給它
    m1 = Len(d(i, 1))             '用len()來計算idno的長度,指派給m1
    If m1 <> 10 Then ec = "FORMAT ERROE" '若m1不是10，則指派"FORMAT ERROR"給ec
    m2 = Mid(idno, 1, 1)          '用mid()函數指派idno的第1個字元給m2
    If m2 < "A" Or m2 > "Z" Then ec = "FORMAT ERROE" '若m2沒介於A~Z之間，則指派"FORMAT ERROR"給ec

    For j = 2 To 10               '用for迴圈進行idno的第2~10字元做檢查
        m3 = Mid(idno, j, 1)      '用mid()函數指派idno的第j個字元給m3
        If m3 < "0" Or m3 > "9" Then ec = "FORMAT ERROE" '檢查m3，若m3沒介於0~9之間，則指派"FORMAT ERROR"給ec
    Next j
End Sub

```

# 性別碼判定：sp2(i)

L1	D1	D2	D3	D4	D5	D6	D7	D8	D9
----	----	----	----	----	----	----	----	----	----

(B) 性別判定：D1 只可為 1 或 2，1 表男性、2 表女性。

## ✖ 變數

- + sex12(身分證號碼第2碼中，1表男性，2表女性)
- + sexmf(每筆資料的第3欄位，m表男性，f表女性)
- + msex(=sex12 & sexmf，"1m"和"2f"表示格式正確，其餘錯誤)

## ✖ 邏輯

- + sex12=mid(d(i,1),2,1)，令sexmf=d(i,3)
- + 判斷msex=sex12 & sexmf是否為1m或2f？
- + 否則令ec="SEX CODE ERROR"

# 副程式sp2(i)的程式碼

```
Sub sp2(ByVal i)
    Dim sex12, sexmf, msex      '宣告sex12(放身分證第2碼),sexmf(放每筆資料的第3個性別欄位欄位),msex(用以檢驗性別代碼是否正確)
    sex12 = Mid(d(i, 1), 2, 1) '從資料中拿第1個欄位，並取用其第2碼，指派給變數sex12
    sexmf = d(i, 3)            '從資料中拿第3個欄位，指派給變數sexmf
    msex = sex12 & sexmf        'msex，並將sex12和sexmf串接在一起指派給它
    If msex <> "1M" And msex <> "2F" Then ec = "SEX CODE ERROR" '若msex的值不是1m或2f，則指派"SEX CODE ERROE"給ec
End Sub
```



# 檢查碼判定：sp3(i)

(C) 檢查辦法：

(a) 字母 L1 由下列表中，找到其代號兩位，令其為 X1、X2。

X1 為十位數，X2 為個位數。

字母	A	B	C	D	E	F	G	H	J	K	L	M	N
代號	10	11	12	13	14	15	16	17	18	19	20	21	22
字母	P	Q	R	S	T	U	V	X	Y	W	Z	I	O
代號	23	24	25	26	27	28	29	30	31	32	33	34	35

(b) 計算方法：

L1	D1	D2	D3	D4	D5	D6	D7	D8	D9
----	----	----	----	----	----	----	----	----	----

$$Y = X1 + 9 \times X2 + 8 \times D1 + 7 \times D2 + 6 \times D3 + 5 \times D4 + 4 \times D5 + 3 \times D6 + 2 \times D7 + D8 + D9$$

如果 Y 能被 10 整除，則表示此身分證號碼正確。



# 檢查碼判定：**sp3(i)**

## ✖ 變數：

- + L1(用mid(d(i,1),1,1)抓資料中第1欄位的第1個字元，並指派給它)
- + S26(依題目將各個英文字依序加以排列，WZIO要背，調到最後面)
- + x12(使用Instr(s26,L1)函數+9，以符合題目中英文字母下方的數字。ex1:A排第1個，+9後變成10。ex2:Z排第24個，+9後變成33。ex3:0排第26個，+9後變成35)
- + X1(依題意為十位數，令m1\10的商數並指派給它)
- + X2(依題意為個位數，令m1 mod 10的餘數並指派給它)

# 程式碼：sp3(i)

```

Sub sp3(ByVal i)
    Dim L1 = Mid(d(i, 1), 1, 1) '用mid(d(i,1),1,1)抓資料中第1欄位的第1個字元，並指派給它
    Dim s26 = "ABCDEFGHJKLMNOPQRSTUVWXYZIO" '依題目將各個英文字依序加以排列，WZIO要背，調到最後面
    Dim x12 = Instr(s26, L1) + 9 '使用Instr(s26,L1)函數+9，以符合題目中英文字母下方的數字
    Dim X1 = x12 \ 10 '依題意為十位數，令m1\10的商數並指派給它
    Dim X2 = x12 Mod 10 '依題意為個位數，令m1 mod 10的餘數並指派給它
    Dim a(10) '宣告陣列A，用來存放題目中D1~D9的數字
    For j = 2 To 10 '用mid(d(i,1),j,1)抓資料中第1欄位的第j個字元，並指派給a(j-1)
        a(j - 1) = Mid(d(i, 1), j, 1)
    Next
    Dim Y = X1 + 9 * X2 + 8 * a(1) + 7 * a(2) + 6 * a(3) + 5 * a(4) + 4 * a(5) + 3 * a(6) + 2 * a(7) + a(8) + a(9)
    If Y Mod 10 <> 0 Then ec = "CHECK SUM ERROR" '依題目公式代入以算出Y值，並檢驗Y是否被10整除，否則指派"CHECK SUM ERROR"給ec
End Sub

```

# 程式結果製表：wdata()

## × 變數

- + dt(DataTable資料型別)：建立資料表格
- + dr(DataRow資料型別)：資料列

## × 方法

- + dt.Columns.Add("ID\_NO")：增加欄位名稱
- + dt.NewRow：增加一個資料列
- + dt.Rows.add(dr)



# wdata()程式碼

```

Sub wdata()
    Dim dt As New DataTable      '宣告1個新的datatable為dt
    Dim dr As DataRow
    dt.Columns.Add("ID_NO")      '為dt增加標題欄位IDNO
    dt.Columns.Add("NAME")       '為dt增加標題欄位NAME
    dt.Columns.Add("SEX")        '為dt增加標題欄位SEX
    dt.Columns.Add("ERROR")      '為dt增加標題欄位ERROR
    For i = 1 To rno             '用迴圈來增加檔案中的資料至表格列
        dr = dt.NewRow           '新增表格列
        dr(0) = d(i, 1)          'dr的欄位0~3分別填入每筆資料的4個欄位
        dr(1) = d(i, 2)
        dr(2) = d(i, 3)
        dr(3) = d(i, 4)
        dt.Rows.Add(dr)          '確認將dr列的內容增加到dt表格內
    Next
    dgv.DataSource = dt          '物件dgv的資料來源設定為dt
    dgv.Sort(dgv.Columns(0), 0)  '依題本的執行結果進行排序
    dgv.Columns(2).Width = 50     '將第2欄的寬度變小，設定成50
    dgv.Columns(3).Width = 150   '將第3欄的寬度變小，設定成150
End Sub

```



# wdata()對照解說

	IDNO ▲	NAME	SEX	ERROR
▶	A102947623	DENNIS	M	
	B12X767544	ISAAC	M	FORMAT ERROE
	B220713002	CATHY	F	
	E221142995	BETTY	F	CHECK SUM ERROR
	F222490168	DONA	F	
	G220977967	ANGEL	F	
	H221930843	ALICE	M	SEX CODE ERROR
	J220374186	DELIA	F	CHECK SUM ERROR
	P220668834	CLAIRE	F	
	V120498032	DARIUS	M	